
Úvod
V tomto článku je prezentována jednoduchá webová aplikace, v níž  se zobrazuje, popřípadě 
aktualizuje databáze v serveru. Aplikace přitom pro obsluhu grafickéno rozhraní vůbec nepoužívá 
jazyk HTML, nýbrž jazyk Flex a ActionScript. Na straně serveru figuruje CGI program napsaný  v 
jazyku ILE/RPG.

Adobe Flex a ActionScript
Jazyk pro programování GUI (grafického uživatelského rozhraní) Adobe Flex se nazývá MXML 
(Macromedia XML). Jazyk Adobe ActionScript je rozšířením jazyka ECMAScript (stejně jako 
JavaScript). Oba jazyky využívají systému Adobe Flash Player. Programování v jazyku Flex je 
umožněno v rámci balíku Flex_SDK_3, který lze volně stáhnout. Program vytvořený v jazyku Flex 
používá běhové prostředí FlashPlayer 9, které je běžně k dispozici v mnoha prohlížečích.

Zdrojový  program zapsaný  v jazyku MXML musí být zkompilován. Výsledkem kompilace je 
soubor typu SWF. Tento soubor je třeba přemístit do adresáře HTTP serveru, odkud jej pak stahuje 
prohlížeč.

Na webové adrese http://www.flex.org lze nalézt vyčerpávající dokumentaci k jazykům Flex i 
ActionScript a dobré příklady s vysvětlivkami.

Zadání aplikace
Aplikace spočívá v údržbě databázového souboru ITEMS870 v systému IBM System i.

Na straně prohlížeče (klienta) komunikuje se serverem program napsaný  v jazyku Flex. Zdrojový 
program je zapsán v souboru DB_Updat.mxml. Ten je zkompilován do souboru DB_Updat.swf 
s použitím kompilátoru mxmlc. Soubor DB_Updat.swf představuje jak uživatelské rozhraní (GUI) 
tak spojení se serverem prostřednictvím třídy HTTPService. Objekt třídy  HTTPService je rozhraní 
mezi prohlížečem a HTTP serverem. Toto rozhraní má tu výhodu, že stránka prohlížeče zůstává 
stále aktivní, zatímco se odesílají a přijímají data. Prohlížeč nemusí čekat na celou novou stránku 
jako odpověď na svůj požadavek (což je jinak běžné v rozhraní HTML), ale zobrazuje jen změněná 
data.

Na straně serveru je protějškem prohlížeče program UPDITEMS, který  je napsán v jazyku ILE/
RPG a je provozován jako CGI skript v prostředí HTTP serveru. Program UPDITEMS využívá 
servisního programu CGISRVPGM2 z balíku CGIDEV2 tak, že volá jeho vybrané procedury. 
Informace o tomto softwarovém balíku lze nalézt na adrese http://www.easy400.net/cgidev2/start.

Postup zpracování dat v aplikaci
Ke spuštění aplikace uživatel zadá adresu v prohlížeči takto:

http//:193.179.195.140:8014/cgivz/DB_Updat.swf

kde 193.179.195.140 je jméno serveru, 8014 číslo portu zvolené pro instanci HTTP serveru. /cgivz 
je adresář, kde je umístěn soubor DB_Updat.swf. Adresář /cgivz je konfigurován v odpovídající 
instanci HTTP serveru.

Poznámka: CGIVZ je také jménem knihovny, kde je umístěn CGI program UPDITEMS (napsaný v 
jazyku RPG) a databázový soubor ITEMS870.

Poté, co prohlížeč stáhl soubor DB_Updat.swf, je vyvolán program UPDITEMS s parametry 
požadavku HTTP. Na základě parametrů program přečte záznamy ze souboru ITEMS870 a pošle je 
do prohlížeče ve formátu XML. Tam je Flex program, tj. soubor DB_Updat.swf, interpretuje a 

zobrazí. Program UPDITEMS využívá soubor XML_ITEMS870.xml jako pomůcku k vytvoření dat 
v XML formátu. Tento soubor slouží jako předloha (šablona) procedurám CGIDEV2.

Data zobrazená v prohlížeči vyhlížejí takto:

Aplikace umožňuje prohlížení (Browse), aktualizaci (Update) a rušení (Delete) záznamů 
v databázovém souboru. Každé tlačítko generuje poněkud jiné parametry požadavku HTTP a 
vyvolá program UPDITEMS. Ten pak pošle zpět  příslušnou odpověď ve tvaru XML. Odpověď se 
skládá z požadovaného počtu záznamů ze souboru ITEMS870.

Kompilace MXML souboru
Ke kompilaci MXML souboru se používá kompilátor mxmlc. Kompilátor mxmlc je soubor 
umístěný  v adresáři bin adresáře Flex_SDK_3. Existují verze pro systémy Windows a unixové 
systémy.

Následující příkazy  kompilují soubor DB_Updat.mxml, za předpokladu, že balík Flex_SDK_3 byl 
instalován do kořenového adresáře a že soubor DB_Updat.mxml je umístěn v adresáři /Flex3/src.
C:\Flex_SDK_3\bin mxmlc  C:\Flex3\src\DB_Updat.mxml

v systémech Windows nebo
/Flex_SDK_3/bin/mxmlc /Flex/src/DB_Updat.mxml

v unixových systémech (např. MacOS X). Výsledný  soubor je DB_Updat.swf a je uložen do 
stejného adresáře /Flex3/src jako zdroj.

Kopírování souboru SWF do serveru
Flex program se kompiluje v osobním počítači. Zkompilovaný  soubor DB_Updat.swf se pak musí 
přenést  do adresáře /cgivz k testování a provozu. Soubor musí být zkopírován jako binární. Lze jej 
přetáhnout myší s použitím Navigátoru iSeries nebo použít program FTP. Tyto procesy lze provádět 
jak v lokální, tak ve vzdálené síti.

Soubor ITEMS870
Databázový soubor ITEMS870 specifikuje klíčové slovo CCSID(870) u pole ITEMDESC ke 
správné interpretaci znaků Latin 2.
     A                                      UNIQUE
     A          R ITEMSR
      *   Item number
     A            ITEMNBR        5          COLHDG('Item' 'number')
      *   Unit price
     A            UNITPR         9  2       COLHDG('Unit' 'price')
      *   Item description
     A            ITEMDESC      50          COLHDG('Item description')
     A                                      CCSID(870)
      *   Key field
     A          K ITEMNBR

Program UPDITEMS
     h dftActGrp(*NO) actGrp(*NEW)                                       (1)
      /copy CGIVZ/QRPGLESRC,HSPECS                                       (2)
      /copy CGIVZ/QRPGLESRC,HSPECSBND

      *================================================================
      *   File definitions
      *================================================================

      *   Data file
     fITEMS870  uf a e           k disk    usropn

      *================================================================
      *   Data definitions
      *================================================================

      /copy CGIDEV2/QRPGLESRC,PROTOTYPEB                                 (3)
      /copy CGIDEV2/QRPGLESRC,USEC
      /copy CGIDEV2/QRPGLESRC,VARIABLES3

     d qrystring       s            190a   varying                       (4)
     d priceChar       s             30a
     d nbrRecords      s             10u 0
     d MAX             c                   const(1000)

      *------------------------------------------------------
      *   variables into which browser request will be copied
      *------------------------------------------------------
       // starting item number to be sent to the browser
     d startItemNbr    s                   like(ITEMNBR)                (5)

       // action to be performed (BROWSE, UPDATE, DELETE)
     d action          s              6a                                (6)

       // variables to be copied into record fields for update or write
     d itemNbr2        s                   like(ITEMNBR)
     d unitPr2         s                   like(UNITPR)
     d itemDesc2       s                   like(ITEMDESC)

      *================================================================
      *   Main procedure
      *================================================================

      /free
       if not %open(ITEMS870);
          open ITEMS870;                                                 (7)
       endif;

       setNoDebug(*off);  // set debugging on                            (8)

       // get input from the browser
       nbrVars = zhbGetInput(qrystring:qusec);                           (9)
       // abort if any errors.
       if qusbavl > 0;

         wrtdebug('Cannot get input data');
       endif;

       // extract starting item number
       startItemNbr = zhbGetVar('startItemNbr');                        (10)

       wrtDebug('startItemNbr: ' + startItemNbr);  // write debug info

       // extract action (BROWSE, UPDATE or DELETE)
       action   = zhbGetVar('action');

       if action = 'BROWSE';  // no other input is needed for browse
          exsr sendFile;      // send records required by the browser
          return;             // return to the server
       endif;

       // extract item number
       itemNbr2 = zhbGetVar('itemNumber');

       // extract unit price
       priceChar = zhbGetVar('unitPrice');
       unitPr2 = C2N2(priceChar);  // convert unit price to numeric

       // extract description
       itemDesc2 = zhbGetVar('itemDescription');

       //   process record obtained from the browser
       //   ----------------------------------------
       chain itemNbr2 ITEMS870;  // read database record by key

       // copy values from browser to database fields
       ITEMNBR  = itemNbr2;
       UNITPR   = unitPr2;
       ITEMDESC = itemDesc2;

       if not %found;   // add the record to the file
          write ITEMSR;

       elseif action = 'UPDATE';  // update the record found
          update ITEMSR;
       elseif action = 'DELETE';  // delete the record found
          delete ITEMSR;
       endif;

       exsr sendFile;  // send data records to the browser

       //  end program without closing
       //  ---------------------------
       return;                                                           (11)

       //==============================================================
       //   Subroutines
       //==============================================================

       //--------------------------------------------------------------
       //   sendFile - send data records to the browser
       //--------------------------------------------------------------

       begsr sendFile;

       clrHtmlBuffer();  //  clear output buffer                         (12)

       //  read XML template into memory (mind the exclamation mark code)
       getHtmlIfs('/cgivz/XML_ITEMS870.xml': '<!-- Sec_': ' -->');       (13)

       //  write XML header section into buffer
       wrtSection('Header');                                             (14)

       // set position in the file to starting item number
       setll startItemNbr ITEMS870;                                      (15)

       // read the first record from the database file - no lock
       read(n) ITEMS870;

       // process max. MAX records of the file
       nbrRecords = 0;
       dow not %eof and nbrRecords <= MAX;
          // transfer database fields to XML value fields
          UpdHTMLVar('itemNbr': ITEMNBR);                                (16)
          UpdHTMLVar('unitPr': %char(UNITPR));
          UpdHTMLVar('itemDescr': ITEMDESC);

          wrtSection('Record');  //  write XML record section            (17)

          // read next record from the database file - no lock
          read(n) ITEMS870;
          nbrRecords += 1;
       enddo;

       wrtSection('Trailer');  // write XML trailer section              (18)

       wrtSection('*fini');  // write output buffer to the browser       (19)

       endsr; // sendFile

(1) Aktivační skupina *NEW je zadána pro snadnější ladění. Pojmenovanou aktivační skupinu není snadné 
ukončit po nové kompilaci (je nutné najít úlohu – job v serveru a ukončit jej). Pojmenovaná aktivační 
skupina by měla být zadána až po odladění pro provoz.

(2) Člen HSPECS obsahuje H příkazy ke kompilaci modulu. Člen HSPECSBND obsahuje H příkazy ke 
spojování.

(3) Člen PROTOTYPEB obsahuje prototypy procedur CGIDEV2. Datová struktura USEC je použita 
v některých procedurách k získání chybových zpráv. Člen VARIABLES3 obsahuje definice pracovních 
proměnných použitých v některých procedurách.

(4) Proměnná qrystring je potřebná pro proceduru zhbGetInput (9), ale není použita, protože je prázdná 
(prohlížeč neposílá požadavek GET).

(5) Proměnná startItemNbr dostává hodnotu z prohlížeče. Do prohlížeče se pak pošlou záznamy počínaje 
tímto číslem položky (item number).

(6) Proměnná action získá hodnotu z prohlížeče. Podle této hodnoty se provede akce Browse, Update nebo 
Delete. Proměnné itemNbr2, unitPr2 a itemDesc2 odpovídají polím z databázového záznamu a 
získávají hodnoty z prohlížeče.

(7) Není-li soubor otevřen, otevře se. Zůstává otevřený, protože program se vrací s vypnutým indikátorem 
LR.

(8) Procedura setNoDebug() povoluje (*OFF) nebo zakazuje (*ON) zápis ladicí informace do souboru 
CGIDEBUG.

(9) Procedura zhbGetInput čte data z požadavku HTTP do interního bufferu a vrací počet proměnných 
(parametrů) z požadavku. Požadavek POST je zadán jako XML structura ve Flex programu (viz Flex 
program níže, bod (4)) a je dále transformován do této podoby:
startItemNbr=00012&action=BROWSE&itemNumber=00023&unitPrice=23.00&itemDescription=Ite
m 00023
Je sestaven z parametrů tvaru proměnná=hodnota oddělených znakem &. Jejich pořadí nemusí být vždy 
stejné. Je-li zjištěna nějaká chyba, nastaví se do podpole qusbavl datové struktury qusec nenulové číslo. 
V tomto případě se zapíše ladicí informace do souboru CGIDEBUG.

(10) Procedura zhbGetVar vyjme a vrátí hodnotu proměnné startItemNbr. Procedura wrtDebug zapíše 
informaci z parametru do souboru CGIDEBUG. Podobně jsou získány zbývající hodnoty proměnných, 
přičemž hodnota proměnné unitPrice je převedena ze znakového tvaru do pakovaného dekadického 
tvaru pomocí procedury C2N2.

(11) Program se vrací s vypnutým indikátorem LR, aby nechal otevřený databázový soubor a statické 
proměnné beze změny pro příští volání.

(12) Podprogram sendFile posílá datové záznamy do prohlížeče. Výstupní data se sestavují v bufferu, který 
je nejprve vyčištěn clrHtmlBuffer.

(13) Procedura getHtmlIfs přečte předlohu XML z adresáře IFS do paměti. Druhý parametr definuje levý 
oddělovač jména sekce, třetí parametr definuje pravý oddělovač. Předloha se používá ke generování 
výstupních dat určených pro prohlížeč.

(14) Procedura wrtSection přidá sekci Header do bufferu.
(15) Operace SETLL nastaví ukazatel na počáteční číslo položky (item number) zjištěné z požadavku 

prohlížeče.
(16) Procedura UpdHtmlVar nahradí parameter v předloze (včetně dvojznaků /% a %/) odpovídající 

hodnotou vzatou z právě přečteného databázového záznamu. Pakovaná nebo binární čísla musí být 
převedena na znaky.

(17) Sekce Record je přidána k bufferu tolikrát, kolikrát je přečten záznam ze souboru, maximálně v počtu 
MAX.

(18) Sekce Trailer je přidána k bufferu jen jednou.
(19) Je-li parametr procedury WrtSection speciální kód *FINI, pošle se obsah bufferu do prohlížeče (jako 

HTTP odpověď) a buffer se vyčistí.

Předloha XML pro CGIDEV2
Ke generaci výstupního proudu tvaru XML, který se posílá do prohlížeče jako HTTP odpověď, 
slouží předloha umístěná v IFS adresáři jako soubor /cgivz/XML_ITEMS870.xml. Podle předlohy 
generují procedury  CGIDEV2 jednotlivé sekce datového proudu do výstupního bufferu. Procedury 
byly původně navrženy ke generaci HTML stránek, ale zde poslouží stejně dobře.

Tvar předlohy je tento:
<!-- Sec_Header -->
Content-Type: text/xml

<?xml version="1.0"?>
<items>
<!-- Sec_Record -->
  <record>
    <itemNumber>/%itemNbr%/</itemNumber>

    <unitPrice>/%unitPr%/</unitPrice>
    <itemDescription>/%itemDescr%/</itemDescription>
  </record>
<!-- Sec_Trailer -->
</items>

Sekce Header je generována jednou, jakož i sekce Trailer. Sekce Record je generována tolikrát, 
kolik se posílá záznamů ze ITEMS870, přičemž se mezi oddělovače /% a %/ dosazují konkrétní 
hodnoty polí.

Soubor předlohy v této aplikaci je zapsán v kódu CCSID 870 a byl konstruován podle pravidel 
doporučených v dokumentaci CGIDEV2. Pozor na vykřičník, ten je velmi silně závislý  na kódování 
znaků (character encoding)! Někdy je také vidět jako svislá čára ( | ), někdy jako pravá hranatá 
závorka (  ] ). Správný hexadecimální kód pro vykřičník v kódu

(1) CCSID 37 je X‘5A‘, což je správný kód, protože soubor QRPGLESRC v knihovně CGIVZ je kódován 
v CCSID 37,

(2) CCSID 870 je X‘4F‘; tento kód je svislá čára ( | ) v CCSID 37.
(3) ASCII Cp819 (iso-8859-1 Latin 1) a Cp1252 (Windows Western Europe Latin 1) je X‘21‘,
(4) ASCII Cp912 (iso-8859-2 Latin 2) je X‘5B‘,

Poznámka: X‘5A‘ v CCSID 870 odpovídá pravé hranaté závorce ( ] ).

Sekce jsou tak uvedeny v komentářovém řádku XML a RPG program se na ně odkazuje jejich 
jmény.

Výsledek generace všech sekcí vyhlíží takto:
Content-Type: text/xml

<?xml version="1.0"?>
<items>
  <record>
    <itemNumber>00001 </itemNumber>
    <unitPrice>1.00 </unitPrice>
    <itemDescription>Item 00001 /</itemDescription>
  </record>
  <record>
    <itemNumber>00002 </itemNumber>
    <unitPrice>2.00 </unitPrice>
    <itemDescription>Item 00002< /itemDescription>
  </record>
  ...
</items>

Tato XML struktura se po přenosu objeví v proměnné itemRecords:XML (viz Flex program níže – 
bod (8)). Kořenový uzel <items> se ve Flex programu nepoužívá, ale ve struktuře musí být 
přítomen, třeba i s jiným jménem. Řádek Content-Type: text/xml je určen pro HTTP protokol a musí 
být následován prázdným řádkem.

Flex program
Následující text  je zdrojový  MXML program DB_Updat.mxml popisující uživatelské rozhraní 
aplikace a způsob komunikace s HTTP serverem.
<?xml version="1.0" encoding="utf-8"?>

<mx:Application                                              (1)

  xmlns:mx="http://www.adobe.com/2006/mxml" 

  creationComplete="init();"                                 (2)

>


  <!-- Define the HTTP service -->

  <mx:HTTPService                                            (3)

    id="itemPrices"

    url=http://193.179.195.140:8014/cgivzp/UPDITEMS.pgm

    method="POST"

    resultFormat="e4x"

    result="resultHandler(event);"

    fault="faultHandler(event);"

  > 

    <mx:request>
                                        (4)

      <startItemNbr>{startItemNbr}</startItemNbr>

      <action>{action}</action>

      <itemNumber>{itemNumberInput.text}</itemNumber>

      <unitPrice>{unitPriceInput.text}</unitPrice>

      <itemDescription>{itemDescriptionInput.text}</itemDescription>

    </mx:request> 

  </mx:HTTPService>




  <mx:Number id="selectionIndex">{myDataGrid.selectedIndex}</mx:Number>      (5)

    

  <mx:Script>                                                                (6)

    <![CDATA[

      import flash.net.navigateToURL;

      import mx.controls.Alert;

      import mx.rpc.events.FaultEvent;

      import mx.rpc.events.ResultEvent;

      private var savedIndex:int; // data grid index saved before "itemPricesUpdate" request   (7)

        [Bindable] private var itemRecords:XML; // database records in XML format              (8)

        [Bindable] private var action:String; // action: BROWSE, UPDATE or DELETE              (9)

        [Bindable] private var startItemNbr:String; // starting item number to be displayed   (10)

      // initial actions

      private function init():void {                                                          (11)

        savedIndex = 0;    // set the first row (as "the row last processed")

        action = 'BROWSE'; // show records

        startItemNbr = ""; // set low value for starting item number

        itemPrices.send(); // send first request to the server to get XML item records        (12)

      }

      // Result handler - gets called once server data is loaded in

      private function resultHandler(event:ResultEvent):void {                                (13)

        itemRecords = event.result as XML;     // fill itemRecords (dataProvider) with data

        myDataGrid.validateNow();              // to scroll correctly; why???

        myDataGrid.scrollToIndex(savedIndex);  // make that row the first in the table

        myDataGrid.selectedIndex = savedIndex; // make saved index selected index

      }

      // Fault handler - displays the error

      private function faultHandler(event:FaultEvent):void {                      (14)

        Alert.show(event.fault.message, "Unable to read server data");

      }

      // Delete function

      private function deleteAction():void {                                      (15)

        // save selected index to set a row near the current one next time

        savedIndex = myDataGrid.selectedIndex;

        action = "DELETE";  // specify delete action

        itemPrices.send();  // send update request with DELETE action to the server

      }

 


    ]]>

10

http://iassist:8014/cgivzp/UPDITEMS.pgm
http://iassist:8014/cgivzp/UPDITEMS.pgm


  </mx:Script>

  <!-- User Interface -->

  <mx:Panel                                                                   (16)

    title="XML Item Prices Update"

    paddingBottom="10" paddingLeft="10" paddingRight="10" paddingTop="10"

    width="100%" height="100%"

  >

    <!-- Display starting item number -->

    <mx:Label

      text="Choose starting item number" fontWeight="bold"

    />

    <mx:TextInput 

      id="startItemNbrInput" 

      width="100"

      text="{startItemNbr}"

      change="startItemNbr = startItemNbrInput.text"                         (17)

    />

    <!-- Display records in a grid -->

    <mx:DataGrid id="myDataGrid" dataProvider="{itemRecords.record}" rowCount="8">        (18)

      <mx:columns>

        <mx:DataGridColumn dataField="itemNumber" headerText="Item Number" width="100"/>

        <mx:DataGridColumn dataField="itemDescription" headerText="Description" width="290"/>

      </mx:columns>

    </mx:DataGrid>

    <!-- Detail view: Display detail record data  -->

    <mx:Form width="100%" autoLayout="false">                                             (19)

      <mx:FormHeading label="Record details"/>

      <mx:FormItem label="Item Number:" width="100%">

        <mx:TextInput 

          id="itemNumberInput" 

          width="100"

          text="{itemRecords.record[selectionIndex].itemNumber}"                          (20)

          change="itemRecords.record[selectionIndex].itemNumber = itemNumberInput.text"   (21)

        />

      </mx:FormItem>

      <mx:FormItem label="Unit Price:" width="100%">

        <mx:TextInput 

          id="unitPriceInput"

          width="150"

          text="{itemRecords.record[selectionIndex].unitPrice}"

          change="itemRecords.record[selectionIndex].unitPrice = unitPriceInput.text"

        />

      </mx:FormItem>

      <mx:FormItem label="Description:" width="100%">

        <mx:TextInput 

          id="itemDescriptionInput"

          width="290"

          text="{itemRecords.record[selectionIndex].itemDescription}"

          change="itemRecords.record[selectionIndex].itemDescription 

                  = itemDescriptionInput.text"

          />

      </mx:FormItem>

    </mx:Form> 

    <!-- Display control bar with buttons -->

    <mx:ControlBar horizontalAlign="center">
                                         (22)

      <mx:Button label="Browse" click="savedIndex = myDataGrid.selectedIndex;         (23)

                 action = 'BROWSE';

                 itemPrices.send();"/>

      <mx:Button label="Update" click="savedIndex = myDataGrid.selectedIndex;         (24)

                 action = 'UPDATE';

                 startItemNbr = (startItemNbr > itemNumberInput.text) ?

                 itemNumberInput.text : startItemNbr;

                 itemPrices.send();"/>

      <mx:Button label="Delete" click="deleteAction();"/>                             (25)

    </mx:ControlBar>
 


  </mx:Panel>


</mx:Application>

Program se skládá ze značek (tags), které jsou označeny předponou mx:. Značky mohou obsahovat 
attributy zvané vlastnosti (properties), které jsou několika druhů: event, style, effect atd. 
ActionScript připomíná jazyk Java v mnoha ohledech. Definuje balíky (packages),  třídy  (classes), 
rozhraní (interfaces), datové členy (data members), konstruktory  (constructors), metody (methods) 
atd. Kromě metod a konstruktorů mohou být definovány funkce (functions) zapsané mimo třídu.

(1) Značka mx:Application obaluje celé grafické rozhraní.
(2) Její vlastnost event - creationComplete zadává funkci init() (11), která se vyvolá okamžitě poté, co je 

vykresleno celé grafické uživatelské rozhraní.
(3) Značka mx:HTTPService definuje způsob komunikace s HTTP serverem. Její jméno je itemPrices a 

obsahuje několik vlastností:
- url zadává adresu programu UPDITEMS, který se má vyvolat prostřednictvím funkce 

itemPrices.send() (12), (15), (23), (24). 
- method=“POST“ říká prohlížeči, aby poslal HTTP serveru požadavek POST.
- resultFormat=“e4x“ (E4X značí ECMAScript For XML) říká, že od serveru (programu UPDITEMS) 

očekává datové záznamy ve formátu XML.
- result zadává funkci resultHandler(event), která se vyvolá, jakmile dorazí všechny datové záznamy 

ze serveru (13).
- fault zadává funkci faultHandler(event), která se vyvolá, když se vyskytne nějaká chyba bránící 

získání dat ze serveru.
(4) Značka mx:request ve značce mx:HTTPService definuje strukturu dat posílaných serveru jakožto data 

hlavičky HTTP pro metodu POST. Jména uzlů startItemNbr a action jsou parametry pro CGI program 
UPDITEMS, které mu sdělují, co má provést. Ostatní jména uzlů odpovídají jménům databázových polí 
a tato jména jsou citována v programu UPDITEMS. Jména proměnných uvnitř uzlů jsou uzavřena ve 
složených závorkách (curly braces), což znamená, že na jejich místa se zapíše obsah z vazebních 
proměnných, kdykoliv tyto změní svou hodnotu. Vazební proměnné (bindable variables) jsou v kódu 
ActionScript označeny metaznačkou  (metadata tag) [Bindable]:
- startItemNbr mění hodnotu, když uživatel zapíše údaj do vstupního pole (17) nebo ve funkci init() 

(11), nebo po klepnutí na tlačítko Update (24),
- action mění hodnotu po klepnutí na tlačítko Browse, Update nebo Delete (23), (24), (15),
- ostatní uzly získají hodnoty, když se změní proměnná itemRecords (8) ve funkci resultHandler() 

(13).
(5) Značka mx:Number označuje číselnou proměnnou selectionIndex pojmenovanou pomocí vlastnosti id. 

Její hodnota se okamžitě kopíruje sem, když uživatel vybere (označí) řádek tabulky myDataGrid. 
V jazyku Flex se grafická tabulka jmenuje data grid (datová síťka, mřížka).

(6) Značka mx:Script vymezuje kód v jazyku ActionScript. Příkazy import dovolují použít nezbytné třídy 
pro definice funkcí.

(7) Proměnná savedIndex je celočíselná a slouží k očíslování vybraného řádku v tabulce (20).
(8) Proměnná itemRecords je typu XML a je plněna ve funkci resultHandler() (13) v rámci značky 

mx:HTTPService (3). Struktura této proměnné je popsána výše v odstavci Předloha XML pro 
CGIDEV2. Proměnná je označena metaznačkou [Bindable], což znamená, že hodnoty všech jejích uzlů 
se automaticky kopírují do míst, kde se jejich jména vyskytují ve složených závorkách (18), (20) aj.

(9) Proměnná action je typu String a je označena metaznačkou [Bindable]. Znamená to, že když se změní 
její obsah, zkopíruje se nová hodnota automaticky do všech míst, kde se její jméno vyskytuje ve 
složených závorkách (4).

(10) Proměnná startItemNbr je označena metaznačkou [Bindable] a je citována ve složených závorkách (4), 
(17), kam se kopíruje její obsah, když se změní.

(11) Funkce init() nejprve vynuluje „uložený“ (saved) index, aby se poprvé zobrazil ve vstupních polích 
první záznam získaný z dodavatele dat (data provider). Nastaví se akce BROWSE (prohlížení 
záznamů), do počátečního čísla položky (starting item number) se dosadí prázdný řetězec (nízká 
hodnota).

(12) Pak se teprve vyvolá metoda itemPrices.send() (která pak volá CGI program UPDITEMS).
(13) Funkce resultHandler()

- přečte data do struktury itemRecords XML, jakmile všechna data dorazí ze serveru (event je  objekt 
typu ResultEvent); proměnná itemRecords.record je definována jako dodavatel dat (data provider) 
do tabulky (18), kam se data automaticky nakopírují,

- ověří data v tabulce metodou myDataGrid.validateNow(), aby se příště při vyvolání funkce řádky 
správně nastavily na požadovaný index; jinak by nastavení nemuselo fungovat  správně.

- volá metodu myDataGrid.scrollToIndex(savedIndex), která zařídí, aby se indexovaný řádek zobrazil 
v tabulce jako první (podobně jako u klíčového slova SFLRCDNBR v DDS).

- dosadí uložený index do proměnné myDataGrid.selectedIndex, což způsobí, že se odpovídající řádek 
označí jako vybraný (označený).

(14) Funkce faultHandler() se vyvolá, když se vyskytne nějaká chyba při přijímání dat ze serveru. Metoda 
Alert.show() zobrazí chybovou zprávu v prohlížeči. 

(15) Funkce deleteAction() řeší klepnutí na tlačítko formou volání funkce. Funkce
- uloží index vybraného řádku tabulky do proměnné savedIndex, aby právě tento záznam se příště 

zobrazil ve vstupních polích,
- nastaví proměnnou action na DELETE, což automaticky naplní uzel <action> ve značce 

<mx:request> (4),
- volá metodu itemPrices.send(), která pak volá CGI program UPDITEMS.

(16) Značka mx:Panel je obsažena ve značce mx:Application.
(17) Značka mx:TextInput s vlastností id=startItemNbrInput (jméno proměnné) je vstupní pole, které 

zobrazuje počáteční číslo položky. Jakmile se změní údaj v poli, např. když uživatel vloží data 
z klávesnice, proměnná startItemNbr (10), která je i vazební, se naplní obsahem tohoto vstupního pole.

(18) Značka mx:DataGrid představuje tabulku (data grid) se dvěma sloupci  8 řádky (včetně záhlaví). Je 
automatickys naplněna daty, když se naplní dodavatel dat (data provider), tj. XML proměnná 
itemRecords (13).

(19) Značka mx:Form obaluje záhlaví a položky formuláře. Položky formuláře obsahují nápisy (labels) a 
vstupní textová pole.

(20) Značka mx:TextInput je vstupní pole, které přijímá data z klávesnice nebo automaticky z XML 
proměnné itemRecords (viz odstavec Předloha XML pro CGIDEV2 výše). Tato proměnná se chová 
jako pole (array) záznamů. Vlastnost text vstupního pole je vázána (uzavřena ve složených závorkách) 
k právě vybranému prvku pole itemRecords.record (indexované proměnnou selectionIndex) a jeho 
podpoli itemNumber.

(21) Když uživatel změní vstupní pole, vyvolá se událost změny (change event) a prvek pole 
itemRecords.record s indexem selectionIndex se naplní změněným textem. Protože proměnná 
itemRecords je vazební (8), je tentýž text uložen také do odpovídajícího řádku tabulky myDataGrid 
(18) a tam také zobrazen.

(22) Značka mx:ControlBar umožňuje umístit řídicí prvky (tlačítka apod.) na dno panelu.
(23) Značka mx:Button zadává vlastnost klepnutí (click property), v níž je zapsáno několik příkazů. Tyto 

příkazy se provedou, když uživatel klepne na tlačítko:
- index právě vybraného řádku tabulky se uloží do proměnné savedIndex pro pozdější použití,
- proměnná action se nastaví na BROWSE, což nastaví také hodnotu <action> v požadavku HTTP (4), 

protože proměnná je vazební (9); ostatní uzly požadavku byly nastaveny automaticky (11), (17),  
(20) atd. již dříve.

- požadavek HTTP je zaslán serveru.
(24) Klepnutí na tlačítko Update  způsobí událost klepnutí (click event), což způsobí provedení několika 

příkazů: mj. se změní počáteční číslo položky (starting item number), aby nově přidaný řádek byl vždy 
viditelný (i kdyby předchozí počáteční číslo položky bylo větší než nové).

(25) Klepnutí na tlačítko Delete vyvolá funkci deleteAction() (15), která provede podobné příkazy jako u 
tlačítka Browse (23). Hodnota proměnné savedIndex bude odpovídat řádku nejbližšímu ke zrušenému 
řádku, což je žádoucí. Tento způsob řešení voláním funkce v jazyku ActionScript je alternativa k zápisu 
série příkazů přímo do vlastnosti klepnutí (click property).

Aplikace jako součást HTML
Aplikace může být také součástí souboru HTML. Skript AC_OETags.js zapsaný  v jazyku JavaScript 
zjišťuje verzi programu Flash Player instalovaného v prohlížeči a je umístěn ve stejném adresáři 
jako HTML soubor. Ten se jmenuje DB_Updat.html a je umístěn v adresáři /cgivz:
<html lang="cz">
  <head>
    <meta http-equiv="Content-Type" content="text/html; charset=utf-8" />
    <title>Adobe Flex 2 - Updating a Database File Using RPG Language and CGIDEV2 Package
    </title>
    <script src="AC_OETags.js " language="javascript"></script>
  </head>

  <body bgcolor="white" text="black" link="blue" alink="green" vlink="fuchsia">
    <h1 align="center">Flex 2 Application Updating a Database File 
        With Conversion From UTF-8 To CCSID 870</h1>
    <h3 align="center">Vladimír Župka</h3>

    <script language="JavaScript" type="text/javascript">
      AC_FL_RunContent(
        "src", "DB_Updat ",
        "width", "70%",
        "height", "65%",
        "align", "left",
        "id", "DB_Updat ",
        "quality", "high",
        "bgcolor", "#869ca7",
        "name", "explorer",
        "allowScriptAccess","sameDomain",
        "type", "application/x-shockwave-flash",
        "pluginspage", "http://www.adobe.com/go/getflashplayer"
      );
    </script>

    <table border=0 align="center" width=300>
      <tr align="left">
        <td> 
          <p>Click on a line in the table and change data in input fields. Then click
             on the <b>Update</b> button below the Panel. If you change <b>Item Number</b> 
             a new record is added into the ITEMS file if the new number is not present 
             in the file; 
             otherwise the existing record will change. If you change another field, data 
             in the current record will be updated.
          <p>Click on the <b>Delete</b> button to delete the record just displayed.
        </td>
      </tr>
    </table>
    <table border=0 align="center" width=300>
      <center>
        <p>
          <img src="Cathedral.JPG" width="200" height="150"</img> 
            <p>Prague Castle - St. Vitus Cathedral
      </center>

    </table>

  </body>
</html>

Internetová adresa, která vyvolá následující obrázek, je 

http:/193.179.195.140:8014/cgivz/DB_Updat.html

Obsah adresáře /cgivz
K provozu Flex aplikace s použitím souboru DB_Updat.swf jsou nutné tyto soubory:
DB_Updat.swf
XML_ITEMS870.xml

K vyvolání aplikace s použitím souboru DB_Updat.html jsou nutné další soubory: 
DB_Updat.html
AC_OETags.js
Cathedral.JPG

Obsah knihovny CGIVZ
Knihovna CGIVZ se skládá z objektů generovaných příkazem CGIDEV2/SETCGILIB. Ostatní 
objekty pro aplikaci jsou:
UPDITEMS - ILE/RPG program

ITEMS870 - fyzický soubor

Obsah adresáře /cgidev
Adresář /cgidev obsahuje soubory a adresáře generované příkazem CGIDEV2/SETCGILIB. 
Adresář /cgidev/conf obsahuje adresář log, kam se zapisují informace o provozu HTTP serveru. 
Obsahuje také soubor httpd.conf, v němž jsou zapsány konfigurační příkazy pro příslušnou instanci 
HTTP serveru. Ostatní objekty nemají žádný vztah ke Flex aplikaci.

Konfigurace HTTP serveru, instance CGIDEV2APA
Hlavní konfigurační příkazy HTTP serveru
ServerRoot /cgidev/conf
DocumentRoot /cgidev/conf/hotdocs
Listen 8014
ThreadsPerChild      53
MaxCGIJobs          100
StartCGI              2    QTMHHTP1
Timeout             120
. . .
Options +ExecCGI -FollowSymLinks -SymLinksIfOwnerMatch +Includes -Indexes +MultiViews
DefaultFsCCSID 870
DefaultNetCCSID 1208
. . .
LanguagePriority cs
<Directory /cgidev/conf/htdocs>
   Order Allow,Deny
   Allow From all
</Directory>
. . .
Alias /cgivz/       /cgivz/
ScriptAliasMatch /cgivzp(.*).pgm /qsys.lib/cgivz.lib/$1.pgm
<Directory /QSYS.LIB/CGIVZ.LIB>
   Options None
   order allow,deny
   allow from all
</Directory>
<Directory /cgivz>
   Options None
   order allow,deny
   allow from all
</Directory>

CGI program běží v úloze CGIDEV2APA typu BCI (batch immediate) pod uživatelským profilem 
QTMHHTP1. Před začátkem testování je potřebí přidělit tomuto profilu oprávnění CHANGE 
k databázovému souboru ITEMS870. Jinak by CGI program při otevírání souboru zhavaroval.

Zpracování znaků v zakódování UTF-8
Flex do sítě posílá a ze sítě přijímá znaková data v kódu UTF-8. Znaky jsou obecně obsaženy 
v proměnném počtu bajtů. Čisté znaky  ASCII (7bitové) jsou kódovány  v jednom bajtu, ostatní 
znaky  typu Western Europe (Latin 1) a Central Europe (Latin 2) potřebují dva bajty a výjimečně tři. 
Např. něměcké písmeno ä (umlaut a) je kódováno jako X’C3A4’.

Ve standardní konfiguraci instance HTTP serveru bude Flex applikace fungovat ve spolupráci 
s programem UPDITEMS čistě, dokud v databázových polích budou jen ASCII znaky. S ostatními 
znaky  bude aplikace stále fungovat, ale znaky v databázových záznamech (nově vytvořených nebo 
přepsaných Flex aplikací) nebudou čitelné na obrazovce 5250, ani jinde.

Je tomu tak proto, že direktivy  HTTP byly nastaveny  na standardní hodnoty. Standardní hodnoty 
překódují požadavky přijaté z prohlížeče z kódu ASCII do EBCDIC. Jiné znaky než  ASCII jsou 
kódovány ve formě tzv. změnových sekvencích (escape sequences). Změnová sekvence má tvar 

%xy kde x a y jsou znakově kódované půlbajty hexadecimální reprezentace bajtu. Např. %C3 je 
změnová sekvence bajtu X’C3‘.

Jiné než ASCII znaky jsou reprezentovány dvěma změnovými sekvencemi odpovídajícími dvěma 
bajtům kódu UTF-8. Např. písmeno ä bude kódováno jako 6 ASCII znaků %C3%A4. Taková 
sekvence se dekóduje v procedurách CGIDEV2 do dvou odpovídajících nesprávných znaků v kódu 
EBCDIC místo jednoho správného znaku.

Instance HTTP serveru v systému System i (V5R3) pro tuto Flex aplikaci byla vytvořena s pomocí 
příkazu CGIDEV2/SETCGILIB z knihovny CGIDEV2. Tento příkaz generuje „provozní“ knihovnu 
CGIVZ. Příkaz také generuje standardní direktivy HTTP serveru. Direktivy HTTP (Apache) byly 
umístěny do souboru /cgidev/conf/httpd.conf. Standardně bylo nastaveno překódování mezi ASCII 
to EBCDIC, což je stejné, jako by byly zadány tyto dvě direktivy:
DefaultFsCCSID 37
  - server character set CCSID 37
DefaultNetCCSID 819
 - network character set CCSID 819

Pro naši aplikaci byly tyto direktivy zaměněny na jiné:
DefaultFsCCSID 870
  - server character set CCSID 870 (Latin 2)
DefaultNetCCSID 1208 - network character set CCSID 1208 (UTF-8)

Tím byla nastavena instance HTTP serveru tak, že překóduje požadavek prohlížeče přímo z kódu 
UTF-8 do kódu CCSID 870 a výstup do prohlížeče opačně. CGI program se pak nemusí zabývat 
rozkódováním změnových sekvencí na vstupu ani jejich vytvářením pro výstup.

Shora zmíněné direktivy mají své protějšky v proměnných prostředí (environment variables) FSCP 
a NETCP. Nastavení proměnných lze ověřit v programu UPDITEMS použitím procedury getenv.
d fsCCSID         s              5a
d netCCSID        s              5a
...
  fsCCSID = getenv ('FSCP': qusec);
  netCCSID = getenv ('NETCP': qusec);
  wrtdebug('fsCCSID: ' + fsCCSID);
  wrtdebug('netCCSID: ' + netCCSID);

Proměnné prostředí lze měnit použitím procedury putenv takto:
  putenv ('FSCP=870': qusec);
  putenv ('NETCP=1208': qusec);

Je to ale pozdě, protože v době, kdy program běží, původní hodnoty proměnných si HTTP server již 
uložil při svém spuštění a nadále je používá. 

Následující obrázek ukazuje, jak Flex zobrazuje data záznamu v textových polích se správými 
diakritickými znaménky.

CL příkaz DSPPFM ukazuje tentýž  záznam souboru ITEMS870 na obrazovce 5250. Je tak 
demonstrována správnost datových převodů:

